# $Id$
# Maintainer: Ray Rashif <schiv@archlinux.org>
# Contributor: Kevin Brubeck Unhammer <unhammer@gmail.com>

pkgname=elyxer
pkgver=1.2.5
pkgrel=1
pkgdesc="A LyX to HTML converter"
url="http://alexfernandez.github.com/elyxer/"
arch=('any')
license=('GPL3')
depends=('python2')
source=("http://alexfernandez.github.com/elyxer/dist/$pkgname-$pkgver.tar.gz")
md5sums=('2692c4c6c5f379c2866eb11eced628bd')

build() {
  cd "$srcdir/$pkgname-$pkgver"

  # python2 fix (copyleftrightupdown Allan McRae 2010)
  sed -i -e "s|#![ ]*/usr/bin/python$|#!/usr/bin/python2|" \
         -e "s|#![ ]*/usr/bin/env python$|#!/usr/bin/env python2|" \
         $(find . -name '*.py')

  # might not be needed, but can't hurt -- kevin
  sed -i "s|('#! /usr/bin/env python')|('#! /usr/bin/env python2')|" \
    src/elyxer/conf/fileconfig.py

  python2 setup.py install --prefix=/usr --root="$pkgdir" --optimize=1
}

# vim:set ts=2 sw=2 et:
